
1 Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 6.1.2.23 which is part of the LTE/SAE test suite in the ‘iwd-TTCN3-B2016-06_D16wk49’ ATS delivery.

The test case can be demonstrated to run one LTE UE (see section 5). Execution logs are provided as evidence.

3 Verification Test Summary

Test Case:
TC_13_5_4
Test Group:
\LTE_A_R12\6_1\Idle_CellReselection_R12.ttcn
ATS Version:
iwd-TTCN3-B2016-06_D16wk49 + essential modifications.

System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
MTK Elbrus
Verification Status:
PASS

4 Corrections required for test case 6.1.2.23
4.1 Change 1
	Function name
	f_TC_6_1_2_23_EUTRA

	Reason for change
	1. Certain TTCN variables which are cell related cannot be invoked before the cell itself is initialised. 
2. When setting the IE freqBandIndicatorPriority_r12, the full structure of corresponding nonCriticalExtension needs to be set. In the current implementation the IE cellAccessRelatedInfo_v1250 would be omitted which is against ASN1 definition.
3. Prior beginning of the test body there is a check of overlapping bands being supported by the UE. However the as a second parameter the wrong pixit is used. 
4. In step 4 where triggering of TAU procedure is expected, a wrong template for RRCConnectionSetupComplete is used.
5. In steps 9 and 10, SRB2 has already been configured, yet the NAS messages are still carried over SRB1.
6. After the test body is finished, UE is in RRC_CONNECTED state, so the postamble should be executed accordingly.


	Summary of change
	1. Security, PLMN, GUTI and LAC variables are invoked after the cell is initialised. 
2. Input parameter in function f_EUTRA_CellInfo_SetSysInfo_SIB1_FBIPrio_r12 has changed to accommodate correct structure of nonCriticalExtension with IE freqBandIndicatorPriority_r12.
(modification of function definition is described in 4.2 Change 2)
3. As a second input parameter into function f_EUTRA_CheckSupportedBand it is used the “px_OverlappingSupportedFrequencyBand_MFBI” instead of “px_OverlappingNotSupportedFrequencyBand_MFBI”.
4. Used generic release 10 template RRCConnectionSetupComplete in step 4 without having the “Rlf_InfoAvailable_r10” set to TRUE.

5. Replacement of SRB1 with SRB2 in steps 9 and 10.

6. Postamble function is called from E2 state instead of E1 state.

	TTCN module
	\LTE_A_R12\6_1\Idle_CellReselection_R12.ttcn

	MCC160 Comment
	1. Accepted
2. Accepted, function renamed to Accepted. Renamed the function to f_EUTRA_CellInfo_SetSysInfo_SIB1v1250_IEs
3. Accepted

4. Accepted
5. Accepted
6. Accepted




Before change

      function f_TC_6_1_2_23_EUTRA ( ) runs on EUTRA_PTC

  { //Inter-band cell reselection / MFBI frequency band priority adjustment / Inter-Band CA

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) InterFreqCarrierFreqList v_InterFreqCarrierList;

    var template (omit) InterFreqCarrierFreqInfo_v9e0_List_Type v_InterFreqCarrierFreqInfo_v9e0_List_Cell1 := omit;

    var template (omit) InterFreqCarrierFreqInfo_v9e0_List_Type v_InterFreqCarrierFreqInfo_v9e0_List_Cell10 := omit;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency_Cell1;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency_Cell10;

    var CellCarrierFreqEUTRA_Type v_CarrierFreqEUTRA_Cell1;

    var CellCarrierFreqEUTRA_Type v_CarrierFreqEUTRA_Cell10;

    var NAS_KsiValue v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType ();

    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType (NORMAL);

    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();

    var GutiParameters_Type v_GutiParams_Cell10:= f_EUTRA_CellInfo_GetGuti(eutra_Cell10);

    var NAS_PlmnId       v_PLMN_Cell10 := f_Asn2Nas_PlmnId(v_GutiParams_Cell10.PLMN_Identity);

    var NAS_Lac          v_LAC_Cell10 := f_EUTRA_CellInfo_GetLocationAreaCode(eutra_Cell10);
    var NAS_MSG_Indication_Type v_NasInd;

    var NasCount_Type v_NasCountUL;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var boolean v_UL_CA := pc_UL_CA;

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell10, -79)

    };

    //Initialise all cell, security and mobile parameters

    f_EUTRA_Init (c3);

    f_EUTRA_CA_InitCells({eutra_Cell1, eutra_Cell10}, CA_InterBand, c3);

    f_EUTRA_CellInfo_SetTAC (eutra_Cell10, tsc_IdleMode_TAC_Cell10);

    //Initialise the Band in SIB1 in FreqBandIndicator and the MFBI frequency

    //Cell 1 uses the E-UTRA Primary frequency band under test (px_ePrimaryFrequencyBand) which together with px_OverlappingSupportedFrequencyBand_MFBI belongs to an E-UTRA CA Configuration for CA Inter-band operation supported by the UE.

    //px_ePrimaryFrequencyBand is different value from px_MFBI_FrequencyBand and px_OverlappingSupportedFrequencyBand_MFBI.

    //Cell 10 belongs to the absolute centre frequency which overlaps between bands controlled by IXITs px_MFBI_FrequencyBand and px_OverlappingSupportedFrequencyBand_MFBI.

    //Cell 1 frequency f1 and Cell 10 frequency f5 are initialised as defined in TS 36.508[18] Table6.2.3.4-1.

    f_EUTRA_InitBandFrequency_MFBI_CA(px_ePrimaryFrequencyBand,

                                      px_ePrimaryBandChannelBandwidth,

                                      f_ConvertDL_BandwidthToUL(px_ePrimaryBandChannelBandwidth),

                                      px_MFBI_FrequencyBand,

                                      px_OverlappingSupportedFrequencyBand_MFBI,

                                      px_MFBI_BandChannelBandwidth,

                                      f_ConvertDL_BandwidthToUL(px_MFBI_BandChannelBandwidth));

    //Initialise SIB1 of cell 10 acc. to Table 6.1.2.23.3.3-1

    f_EUTRA_CellInfo_SetSysInfo_SIB1_FBIPrio_r12(eutra_Cell10, true_);
    //Initialise SIB2 of cell 10 acc. to Table 6.1.2.23.3.3-1a

    f_EUTRA_CellInfo_SetSysInfo_SIB2_v8h0_IEs(eutra_Cell10, cs_SystemInformationBlockType2_v8h0_IEs({1}));

    //Set SIB5 of cell 1 according to Table 6.1.2.23.3.3-1b

    v_ChannelBandwidthDependency_Cell1 := f_EUTRA_BandDependentParam( px_ePrimaryBandChannelBandwidth,

                                                                      f_ConvertDL_BandwidthToUL(px_ePrimaryBandChannelBandwidth));

    v_ChannelBandwidthDependency_Cell10 := f_EUTRA_BandDependentParam( px_MFBI_BandChannelBandwidth,

                                                                       f_ConvertDL_BandwidthToUL(px_MFBI_BandChannelBandwidth));

    v_CarrierFreqEUTRA_Cell10 := f_EUTRA_CellInfo_GetEARFCN(eutra_Cell10);

    v_InterFreqCarrierList := {cs_InterFreqCarrierFreq(v_CarrierFreqEUTRA_Cell10.dl_CarrierFreq, v_ChannelBandwidthDependency_Cell10.AllowedMeasBandwidth)};

    if (ispresent(v_CarrierFreqEUTRA_Cell10.dl_CarrierFreq_v9e0)){

      v_InterFreqCarrierFreqInfo_v9e0_List_Cell10 := {cs_InterFreqCarrierFreq_v9e0(v_CarrierFreqEUTRA_Cell10.dl_CarrierFreq_v9e0, cs_MultiBandInfoList1_v9e0(px_OverlappingSupportedFrequencyBand_MFBI))};

      f_EUTRA_CellInfo_SetSIB5(eutra_Cell1, cs_508_SystemInformationBlockType5_Def(v_InterFreqCarrierList,v_InterFreqCarrierFreqInfo_v9e0_List_Cell10));

      f_EUTRA_CellInfo_SetSysInfo_SIB5_v8h0_IEs(eutra_Cell1,cs_SystemInformationBlockType5_v8h0_IEs({cs_InterFreqCarrierFreqInfo_v8h0({maxFBI})}));

      }

    else{

      f_EUTRA_CellInfo_SetSIB5(eutra_Cell1, cs_508_SystemInformationBlockType5_Def(v_InterFreqCarrierList,v_InterFreqCarrierFreqInfo_v9e0_List_Cell10));

      f_EUTRA_CellInfo_SetSysInfo_SIB5_v8h0_IEs(eutra_Cell1,cs_SystemInformationBlockType5_v8h0_IEs({cs_InterFreqCarrierFreqInfo_v8h0({px_OverlappingSupportedFrequencyBand_MFBI})}));

    }

    //Set SIB5 of cell 10 according to Table 6.1.2.23.3.3-1b

    v_CarrierFreqEUTRA_Cell1 := f_EUTRA_CellInfo_GetEARFCN(eutra_Cell1);

    v_InterFreqCarrierList := {cs_InterFreqCarrierFreq(v_CarrierFreqEUTRA_Cell1.dl_CarrierFreq, v_ChannelBandwidthDependency_Cell1.AllowedMeasBandwidth)};

    if(ispresent(v_CarrierFreqEUTRA_Cell1.dl_CarrierFreq_v9e0)){

      v_InterFreqCarrierFreqInfo_v9e0_List_Cell1 := {cs_InterFreqCarrierFreq_v9e0(v_CarrierFreqEUTRA_Cell1.dl_CarrierFreq_v9e0)};

      }

    f_EUTRA_CellInfo_SetSIB5 (eutra_Cell10, cs_508_SystemInformationBlockType5_Def(v_InterFreqCarrierList, v_InterFreqCarrierFreqInfo_v9e0_List_Cell1));

    //Set maximum cell powel level for cell 10

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell10, -79);

    //Create and configure all cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell10);

    //Bring UE to initial state

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);

    // Check that both px_MFBI_FrequencyBand and px_OverlappingSupportedFrequencyBand_MFBI band are supported by UE

    f_EUTRA_CheckSupportedBand( px_MFBI_FrequencyBand, px_OverlappingNotSupportedFrequencyBand_MFBI );
    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //The SS changes the cells power level setting according to the row "T1" in table 6.1.2.23.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    //The UE transmits an RRCConnectionRequest message on cell 10.

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell10);

    //@siclog "Step 3" siclog@

    // The SS transmits an RRCConnectionSetup message on Cell 10.

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_Cell10, -, tsc_RRC_TI_Def);

    //@siclog "Step 4" siclog@

    // The UE transmits an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment

    // and a TRACKING AREA UPDATE REQUEST message is sent to update the registration of the actual tracking area.

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell10,

                                                       cr_RRCConnectionSetupComplete_r10 (tsc_RRC_TI_Def, ?, -, -, -, true_),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            v_KsiValue,
                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    //@siclog "Step 5 - 6" siclog@

    // The SS transmits a SecurityModeCommand message.

    // The UE transmits a SecurityModeComplete message.

    v_AuthParams := f_EUTRA_RRC_ActivateSecurity ( eutra_Cell10, v_AuthParams, v_NasCountUL );

    f_EUTRA_Security_Set ( v_AuthParams );

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell10);

    //@siclog "Step 7" siclog@

    // The SS transmits an RRCConnectionReconfiguration message to configure data radio bearer.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell10,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2({cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1)},//TBD

                                                                                                                 cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                 cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                            v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                            f_EUTRA_CellInfo_GetAntennaInfoDedicated(eutra_Cell10))),

                                                                        omit)));

    //@siclog "Step 8" siclog@

    // The UE transmits an RRCConnectionReconfigurationComplete message to confirm the establishment of data radio bearer.

    SRB.receive (car_SRB1_RrcPdu_IND(eutra_Cell10, cr_508_RRCConnectionReconfigurationComplete (tsc_RRC_TI_Def)));

    //@siclog "Step 9" siclog@

    // SS responds with TRACKING AREA UPDATE ACCEPT message.

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell10,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(f_GetEPSTAUType(NORMAL),

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams_Cell10),

                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN_Cell10 , {bit2oct(tsc_IdleMode_TAC_Cell10)}),

                                                                  v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus,

                                                                  f_GetLAI (v_PLMN_Cell10, v_LAC_Cell10, NORMAL),

                                                                  f_GetMSId (NORMAL),

                                                                  f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    //@siclog "Step 10" siclog@

    // The UE send a TRACKING AREA UPDATE COMPLETE on the cell 10.

    SRB.receive (car_SRB1_NasPdu_IND(eutra_Cell10,

                                     cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                       cr_508_TAU_Complete)));

    //@siclog "Step 11 - 12" siclog@

    //The SS transmits an RRCConnectionReconfiguration message containing a sCellToAddModList with SCell Cell 1 addition.

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?

    f_EUTRA_508RRC_AddMod_1Scell(eutra_Cell10,

                                 eutra_Cell1,

                                 1,  //SCellIndex_r10

                                 v_UL_CA , // UL CA desired

                                 -,  // C_RNTI     default        := tsc_C_RNTI_Def,

                                 omit,

                                 -,-,-,-,-,-,-,-, infinity_);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.2.23 Step 12");

    f_EUTRA_TestBody_Set(false);

    //Switch/power off UE

    f_EUTRA_Postamble(eutra_Cell10, E1_IDLE);

  }
After change
        function f_TC_6_1_2_23_EUTRA ( ) runs on EUTRA_PTC

  { //Inter-band cell reselection / MFBI frequency band priority adjustment / Inter-Band CA

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) InterFreqCarrierFreqList v_InterFreqCarrierList;

    var template (omit) InterFreqCarrierFreqInfo_v9e0_List_Type v_InterFreqCarrierFreqInfo_v9e0_List_Cell1 := omit;

    var template (omit) InterFreqCarrierFreqInfo_v9e0_List_Type v_InterFreqCarrierFreqInfo_v9e0_List_Cell10 := omit;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency_Cell1;

    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency_Cell10;

    var CellCarrierFreqEUTRA_Type v_CarrierFreqEUTRA_Cell1;

    var CellCarrierFreqEUTRA_Type v_CarrierFreqEUTRA_Cell10;

    //var NAS_KsiValue v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType ();

    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType (NORMAL);

    var EUTRA_SecurityParams_Type v_AuthParams;

    var GutiParameters_Type v_GutiParams_Cell10;

    var NAS_PlmnId       v_PLMN_Cell10;

    var NAS_Lac          v_LAC_Cell10;

    var NAS_MSG_Indication_Type v_NasInd;

    var NasCount_Type v_NasCountUL;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var boolean v_UL_CA := pc_UL_CA;

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell10, -79)

    };

    //Initialise all cell, security and mobile parameters

    f_EUTRA_Init (c3);

    f_EUTRA_CA_InitCells({eutra_Cell1, eutra_Cell10}, CA_InterBand, c3); 

    f_EUTRA_CellInfo_SetTAC (eutra_Cell10, tsc_IdleMode_TAC_Cell10);

    v_GutiParams_Cell10:= f_EUTRA_CellInfo_GetGuti(eutra_Cell10);

    v_PLMN_Cell10 := f_Asn2Nas_PlmnId(v_GutiParams_Cell10.PLMN_Identity);

    v_LAC_Cell10 := f_EUTRA_CellInfo_GetLocationAreaCode(eutra_Cell10);
    //Initialise the Band in SIB1 in FreqBandIndicator and the MFBI frequency

    //Cell 1 uses the E-UTRA Primary frequency band under test (px_ePrimaryFrequencyBand) which together with px_OverlappingSupportedFrequencyBand_MFBI belongs to an E-UTRA CA Configuration for CA Inter-band operation supported by the UE.

    //px_ePrimaryFrequencyBand is different value from px_MFBI_FrequencyBand and px_OverlappingSupportedFrequencyBand_MFBI.

    //Cell 10 belongs to the absolute centre frequency which overlaps between bands controlled by IXITs px_MFBI_FrequencyBand and px_OverlappingSupportedFrequencyBand_MFBI.

    //Cell 1 frequency f1 and Cell 10 frequency f5 are initialised as defined in TS 36.508[18] Table6.2.3.4-1.

    f_EUTRA_InitBandFrequency_MFBI_CA(px_ePrimaryFrequencyBand,

                                      px_ePrimaryBandChannelBandwidth,

                                      f_ConvertDL_BandwidthToUL(px_ePrimaryBandChannelBandwidth),

                                      px_MFBI_FrequencyBand,

                                      px_OverlappingSupportedFrequencyBand_MFBI,

                                      px_MFBI_BandChannelBandwidth,

                                      f_ConvertDL_BandwidthToUL(px_MFBI_BandChannelBandwidth));

    //Initialise SIB1 of cell 10 acc. to Table 6.1.2.23.3.3-1

    f_EUTRA_CellInfo_SetSysInfo_SIB1_FBIPrio_r12(eutra_Cell10, cs_SystemInformationBlockType1_v1250_FBIpriority_r12);
    //Initialise SIB2 of cell 10 acc. to Table 6.1.2.23.3.3-1a

    f_EUTRA_CellInfo_SetSysInfo_SIB2_v8h0_IEs(eutra_Cell10, cs_SystemInformationBlockType2_v8h0_IEs({1}));

    //Set SIB5 of cell 1 according to Table 6.1.2.23.3.3-1b

    v_ChannelBandwidthDependency_Cell1 := f_EUTRA_BandDependentParam( px_ePrimaryBandChannelBandwidth,

                                                                      f_ConvertDL_BandwidthToUL(px_ePrimaryBandChannelBandwidth));

    v_ChannelBandwidthDependency_Cell10 := f_EUTRA_BandDependentParam( px_MFBI_BandChannelBandwidth,

                                                                       f_ConvertDL_BandwidthToUL(px_MFBI_BandChannelBandwidth));

    v_CarrierFreqEUTRA_Cell10 := f_EUTRA_CellInfo_GetEARFCN(eutra_Cell10);

    v_InterFreqCarrierList := {cs_InterFreqCarrierFreq(v_CarrierFreqEUTRA_Cell10.dl_CarrierFreq, v_ChannelBandwidthDependency_Cell10.AllowedMeasBandwidth)};

    if (ispresent(v_CarrierFreqEUTRA_Cell10.dl_CarrierFreq_v9e0)){

      v_InterFreqCarrierFreqInfo_v9e0_List_Cell10 := {cs_InterFreqCarrierFreq_v9e0(v_CarrierFreqEUTRA_Cell10.dl_CarrierFreq_v9e0, cs_MultiBandInfoList1_v9e0(px_OverlappingSupportedFrequencyBand_MFBI))};

      f_EUTRA_CellInfo_SetSIB5(eutra_Cell1, cs_508_SystemInformationBlockType5_Def(v_InterFreqCarrierList,v_InterFreqCarrierFreqInfo_v9e0_List_Cell10));

      f_EUTRA_CellInfo_SetSysInfo_SIB5_v8h0_IEs(eutra_Cell1,cs_SystemInformationBlockType5_v8h0_IEs({cs_InterFreqCarrierFreqInfo_v8h0({maxFBI})}));

      }

    else{

      f_EUTRA_CellInfo_SetSIB5(eutra_Cell1, cs_508_SystemInformationBlockType5_Def(v_InterFreqCarrierList,v_InterFreqCarrierFreqInfo_v9e0_List_Cell10));

      f_EUTRA_CellInfo_SetSysInfo_SIB5_v8h0_IEs(eutra_Cell1,cs_SystemInformationBlockType5_v8h0_IEs({cs_InterFreqCarrierFreqInfo_v8h0({px_OverlappingSupportedFrequencyBand_MFBI})}));

    }

    //Set SIB5 of cell 10 according to Table 6.1.2.23.3.3-1b

    v_CarrierFreqEUTRA_Cell1 := f_EUTRA_CellInfo_GetEARFCN(eutra_Cell1);

    v_InterFreqCarrierList := {cs_InterFreqCarrierFreq(v_CarrierFreqEUTRA_Cell1.dl_CarrierFreq, v_ChannelBandwidthDependency_Cell1.AllowedMeasBandwidth)};

    if(ispresent(v_CarrierFreqEUTRA_Cell1.dl_CarrierFreq_v9e0)){

      v_InterFreqCarrierFreqInfo_v9e0_List_Cell1 := {cs_InterFreqCarrierFreq_v9e0(v_CarrierFreqEUTRA_Cell1.dl_CarrierFreq_v9e0)};

      }

    f_EUTRA_CellInfo_SetSIB5 (eutra_Cell10, cs_508_SystemInformationBlockType5_Def(v_InterFreqCarrierList, v_InterFreqCarrierFreqInfo_v9e0_List_Cell1));

    //Set maximum cell powel level for cell 10

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell10, -79);

    //Create and configure all cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell10);

    //Bring UE to initial state

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);

    v_AuthParams := f_EUTRA_Security_Get();

    // Check that both px_MFBI_FrequencyBand and px_OverlappingSupportedFrequencyBand_MFBI band are supported by UE

    f_EUTRA_CheckSupportedBand( px_MFBI_FrequencyBand, px_OverlappingSupportedFrequencyBand_MFBI );
    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //The SS changes the cells power level setting according to the row "T1" in table 6.1.2.23.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    //The UE transmits an RRCConnectionRequest message on cell 10.

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell10);

    //@siclog "Step 3" siclog@

    // The SS transmits an RRCConnectionSetup message on Cell 10.

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_Cell10, -, tsc_RRC_TI_Def);

    //@siclog "Step 4" siclog@

    // The UE transmits an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment

    // and a TRACKING AREA UPDATE REQUEST message is sent to update the registration of the actual tracking area.

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell10,

                                                       cr_RRCConnectionSetupComplete_r10 (tsc_RRC_TI_Def, ?), 

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            f_EUTRA_SecurityKSIasme_Get(),
                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    //@siclog "Step 5 - 6" siclog@

    // The SS transmits a SecurityModeCommand message.

    // The UE transmits a SecurityModeComplete message.

    v_AuthParams := f_EUTRA_RRC_ActivateSecurity ( eutra_Cell10, v_AuthParams, v_NasCountUL );

    f_EUTRA_Security_Set ( v_AuthParams );

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info(eutra_Cell10);

    //@siclog "Step 7" siclog@

    // The SS transmits an RRCConnectionReconfiguration message to configure data radio bearer.

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell10,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2({cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1)},//TBD

                                                                                                                 cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                 cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                            v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                            f_EUTRA_CellInfo_GetAntennaInfoDedicated(eutra_Cell10))),

                                                                        omit)));

    //@siclog "Step 8" siclog@

    // The UE transmits an RRCConnectionReconfigurationComplete message to confirm the establishment of data radio bearer.

    SRB.receive (car_SRB1_RrcPdu_IND(eutra_Cell10, cr_508_RRCConnectionReconfigurationComplete (tsc_RRC_TI_Def)));

    //@siclog "Step 9" siclog@

    // SS responds with TRACKING AREA UPDATE ACCEPT message.

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_Cell10,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(f_GetEPSTAUType(NORMAL),

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams_Cell10),

                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN_Cell10 , {bit2oct(tsc_IdleMode_TAC_Cell10)}),

                                                                  v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus,

                                                                  f_GetLAI (v_PLMN_Cell10, v_LAC_Cell10, NORMAL),

                                                                  f_GetMSId (NORMAL),

                                                                  f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)))));

    //@siclog "Step 10" siclog@

    // The UE send a TRACKING AREA UPDATE COMPLETE on the cell 10.

    SRB.receive (car_SRB2_NasPdu_IND(eutra_Cell10,

                                     cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                       cr_508_TAU_Complete)));

    //@siclog "Step 11 - 12" siclog@

    //The SS transmits an RRCConnectionReconfiguration message containing a sCellToAddModList with SCell Cell 1 addition.

    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?

    f_EUTRA_508RRC_AddMod_1Scell(eutra_Cell10,

                                 eutra_Cell1,

                                 1,  //SCellIndex_r10

                                 v_UL_CA , // UL CA desired

                                 -,  // C_RNTI     default        := tsc_C_RNTI_Def,

                                 omit,

                                 -,-,-,-,-,-,-,-, infinity_);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.1.2.23 Step 12");

    f_EUTRA_TestBody_Set(false);

    //Switch/power off UE

    f_EUTRA_Postamble(eutra_Cell10, E2_CONNECTED);
  } 
4.2 Change 2
	Function name
	f_EUTRA_CellInfo_SetSysInfo_SIB1_FBIPrio_r12

	Reason for change
	When setting the IE freqBandIndicatorPriority_r12, the full structure of corresponding nonCriticalExtension needs to be set. In the current implementation the IE cellAccessRelatedInfo_v1250 would be omitted which is against ASN1 definition.



	Summary of change
	Input parameter in function f_EUTRA_CellInfo_SetSysInfo_SIB1_FBIPrio_r12 has changed to accommodate correct structure of nonCriticalExtension with IE freqBandIndicatorPriority_r12.

New template with such structure was created – template defined in 4.3 Change 3

	TTCN module
	\LTE_A_R12\6_1\Idle_CellReselection_R12.ttcn

	MCC160 Comment
	Accepted. Renamed the function to f_EUTRA_CellInfo_SetSysInfo_SIB1v1250_IEs


Before change

      function f_EUTRA_CellInfo_SetSysInfo_SIB1_FBIPrio_r12(EUTRA_CellId_Type p_CellId,

                                                        template(omit) SystemInformationBlockType1_v1250_IEs.freqBandIndicatorPriority_r12 p_FreqBandIndicatorPriority_r12) runs on EUTRA_PTC

  {

    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);

    v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.freqBandIndicatorPriority_r12 := p_FreqBandIndicatorPriority_r12;

    f_EUTRA_CellInfo_Set(p_CellId, v_CellInfo);

  }
After change
         function f_EUTRA_CellInfo_SetSysInfo_SIB1_FBIPrio_r12(EUTRA_CellId_Type p_CellId,

                                                        template(omit) SystemInformationBlockType1_v1250_IEs p_FreqBandIndicatorPriority_r12) runs on EUTRA_PTC

  {

    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);

    v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension := p_FreqBandIndicatorPriority_r12;

    f_EUTRA_CellInfo_Set(p_CellId, v_CellInfo);

  }
4.3 Change 3
	Template name
	cs_SystemInformationBlockType1_v1250_FBIpriority_r12

	Reason for change
	Used in 4.2 Change 2

	Summary of change
	New template definition

	TTCN module
	EUTRA_SysInfo_Templates.ttcn

	MCC160 Comment
	Accepted


After change
         template(value) SystemInformationBlockType1_v1250_IEs cs_SystemInformationBlockType1_v1250_FBIpriority_r12 :=

  { 

    cellAccessRelatedInfo_v1250 := {

      category0Allowed_r12 := omit

    },

    cellSelectionInfo_v1250 := omit,

    freqBandIndicatorPriority_r12 := true_,

    nonCriticalExtension := omit

  }
5 Execution Log Files

5.1 MediaTek Elbrus
The MTK Elbrus UE passed this test case on Anritsu Protocol Conformance Test System ME7832L on LTE TDD bands 39-41.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_6_1_2_23_Log.html
6 References
	[1]
	R5s170xxx: Supporting information for agreement of Rel-12 MFBI frequency band priority adjustment/inter-band CA test 6.1.2.23. This archive comprises html and xml format execution log files and PICS/PIXIT settings file respectively

	
	


